

# NAG Fortran Library Routine Document

## G01JDF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G01JDF calculates the lower tail probability for a linear combination of (central)  $\chi^2$  variables.

### 2 Specification

```
SUBROUTINE G01JDF(METHOD, N, RLAM, D, C, PROB, WORK, IFAIL)
INTEGER          N, IFAIL
real           RLAM(N), D, C, PROB, WORK(N+1)
CHARACTER*1     METHOD
```

### 3 Description

Let  $u_1, u_2, \dots, u_n$  be independent Normal variables with mean zero and unit variance, so that  $u_1^2, u_2^2, \dots, u_n^2$  have independent  $\chi^2$  distributions with unit degrees of freedom. G01JDF evaluates the probability that

$$\lambda_1 u_1^2 + \lambda_2 u_2^2 + \dots + \lambda_n u_n^2 < d(u_1^2 + u_2^2 + \dots + u_n^2) + c.$$

If  $c = 0.0$  this is equivalent to the probability that

$$\frac{\lambda_1 u_1^2 + \lambda_2 u_2^2 + \dots + \lambda_n u_n^2}{u_1^2 + u_2^2 + \dots + u_n^2} < d.$$

Alternatively let

$$\lambda_i^* = \lambda_i - d, \quad \text{for } i = 1, 2, \dots, n,$$

then G01JDF returns the probability that

$$\lambda_1^* u_1^2 + \lambda_2^* u_2^2 + \dots + \lambda_n^* u_n^2 < c.$$

Two methods are available. One due to Pan (1964) (see Farebrother (1980)) makes use of series approximations. The other method due to Imhof (1961) reduces the problem to a one-dimensional integral. If  $n \geq 6$  the method described in D01BDF is used to compute the value of the integral otherwise D01AJF is used.

Pan's procedure can only be used if the  $\lambda_i^*$  are sufficiently distinct; G01JDF requires the  $\lambda_i^*$  to be at least 1% distinct; see Section 8. If the  $\lambda_i^*$  are at least 1% distinct and  $n \leq 60$ , then Pan's procedure is recommended; otherwise Imhof's procedure is recommended.

### 4 References

Farebrother R W (1980) Algorithm AS 153. Pan's procedure for the tail probabilities of the Durbin–Watson statistic *Appl. Statist.* **29** 224–227

Imhof J P (1961) Computing the distribution of quadratic forms in Normal variables *Biometrika* **48** 419–426

Pan Jie-Jian (1964) Distributions of the noncircular serial correlation coefficients *Shuxue Jinzhan* **7** 328–337

## 5 Parameters

- 1: METHOD – CHARACTER\*1 *Input*  
*On entry:* indicates whether Pan's, Imhof's or an appropriately selected procedure is to be used.  
 If METHOD = 'P', then Pan's method is used.  
 If METHOD = 'I', then Imhof's method is used.  
 If METHOD = 'D', then Pan's method is used if  $\lambda_i^*$ , for  $i = 1, 2, \dots, n$  are at least 1% distinct and  $n \leq 60$ ; otherwise Imhof's method is used.  
*Constraint:* METHOD = 'P', 'I' or 'D'.
- 2: N – INTEGER *Input*  
*On entry:* the number of independent standard Normal variates, (central  $\chi^2$  variates),  $n$ .  
*Constraint:*  $N \geq 1$ .
- 3: RLAM(N) – *real* array *Input*  
*On entry:* the weights,  $\lambda_i$ , for  $i = 1, 2, \dots, n$ , of the central  $\chi^2$  variables.  
*Constraint:*  $RLAM(i) \neq D$  for at least one  $i$ . If METHOD = 'P', then the  $\lambda_i^*$ , for  $i = 1, 2, \dots, n$ , must be at least 1% distinct; see Section 8.
- 4: D – *real* *Input*  
*On entry:* the multiplier of the central  $\chi^2$  variables,  $d$ .  
*Constraint:*  $D \geq 0.0$ .
- 5: C – *real* *Input*  
*On entry:* the value of the constant,  $c$ .
- 6: PROB – *real* *Output*  
*On exit:* the lower tail probability for the linear combination of central  $\chi^2$  variables.
- 7: WORK(N+1) – *real* array *Workspace*
- 8: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 1$ ,  
 or  $D < 0.0$ ,

or METHOD  $\neq$  'P', 'I' or 'D'.

IFAIL = 2

On entry, RLAM( $i$ ) = D for all values of  $i$ , for  $i = 1, 2, \dots, n$ .

IFAIL = 3

On entry, METHOD='P' yet two successive values of the ordered  $\lambda_i^*$ , for  $i = 1, 2, \dots, n$ , were not at least 1% distinct.

## 7 Accuracy

On successful exit at least four decimal places of accuracy should be achieved.

## 8 Further Comments

Pan's procedure can only work if the  $\lambda_i^*$  are sufficiently distinct. G01JDF uses the check  $|w_j - w_{j-1}| \geq 0.01 \times \max(|w_j|, |w_{j-1}|)$ , where the  $w_j$  are the ordered non-zero values of  $\lambda_i^*$ .

For the situation when all the  $\lambda_i$  are positive G01JCF may be used. If the probabilities required are for the Durbin–Watson test, then the bounds for the probabilities are given by G01EPF.

## 9 Example

For  $n = 10$ , the choice of method, values of  $c$  and  $d$  and the  $\lambda_i$  are input and the probabilities computed and printed.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G01JDF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5, NOUT=6)
INTEGER         N
PARAMETER       (N=10)
*      .. Local Scalars ..
real           C, D, PROB
INTEGER         I, IFAIL
CHARACTER       METHOD
*      .. Local Arrays ..
real          RLAM(N), WORK(N+1)
*      .. External Subroutines ..
EXTERNAL        G01JDF
*      .. Executable Statements ..
WRITE (NOUT,*) 'G01JDF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) METHOD, D, C
READ (NIN,*) (RLAM(I), I=1, N)
*
IFAIL = 0
*
CALL G01JDF(METHOD, N, RLAM, D, C, PROB, WORK, IFAIL)
*
WRITE (NOUT,*)
WRITE (NOUT,99999) ' Values of lambda ', (RLAM(I), I=1, N)
WRITE (NOUT,99999) ' Value of D      ', D
WRITE (NOUT,99999) ' value of C      ', C
WRITE (NOUT,*)
WRITE (NOUT,99998) ' Probability = ', PROB
```

```
      STOP
*
99999 FORMAT (1X,A,10F6.2)
99998 FORMAT (1X,A,F10.4)
      END
```

## 9.2 Program Data

```
G01JDF Example Program Data
'P' 1.0 0.0
-9.0 -7.0 -5.0 -3.0 -1.0 2.0 4.0 6.0 8.0 10.0
```

## 9.3 Program Results

```
G01JDF Example Program Results

Values of lambda  -9.00 -7.00 -5.00 -3.00 -1.00  2.00  4.00  6.00  8.00 10.00
Value of D        1.00
value of C        0.00

Probability =      0.5749
```

---